Here is the complete documentation for this Wideband Magnetic Loop Model implemented in LTSpice.
- 73 John VE6EY

Wideband Magnetic Loop

Equivalent Model Description

John VE6EY
Making It Up
November 2018

Prototype Magnetic Loop Assembly

Here is my first prototype 1 meter diameter loop, made from ½ inch PEX.

Bottom and top braces 3D printed. Both are designed to hold the circular shape of the loop. The bottom brace also has room for loop amplifier. A vertical brace made of ½ inch PVC pipe was also inserted.

Rather than using coax, the loop connections will be made with CAT7 cable. At the lower end, you can see a RJ-45 breakout terminal strip, which enables screw down connections for each of the 8 wires (four pairs) in the CAT cable.

Weight is around 1.2 pounds and wind loading is about 0.5 square foot.

Prototype Mag Loop Baseline Data

My objective is to build a SPICE model of the loop antenna, and then use that model to understand loop performance and design an overall wideband active loop system.

So, I began by taking some raw measurements of loop impedance using my RigExpert AA-30 antenna analyzer. This was done by sweeping, and taking more detailed spot measurements across the 100 kHz to 30 MHz intended range.

Generally, loop resistance is very low, typically 0.5 , except where there is some sort of resonance around 9 MHz. This is caused by the combination of loop inductance and distributed capacitance.

These measurement of loop impedance provide a baseline. Later, if my SPICE model is working correctly, it should produce similar results, in particular the actual self-resonance around 9 MHz.

As things presently stand, this prototype is useless as a wideband receiving antenna when operating in open circuit mode.



Equivalent Circuit

A SPICE model for a loop antenna needs an equivalent circuit that captures the resistance, inductance and capacitance of the mechanical structure.

To perform properly, the equivalent circuit must take into account three types of resistance. DC loss is the simplest, it's just the resistance of the conductor. Both AC loss (skin effect) and radiation resistance are more complex, as they vary according to frequency.

Inductance depends on the parameters of the loop, in particular material, loop and conductor diameters, and number of turns. In addition, stay inductance arises from any length of wire, regardless of shape.

Finally, any loop or conductor generates stray or distributed capacitance which is perhaps unintended but real. Intended capacitance may be added to resonate the loop purposely. Otherwise, the stray capacitance creates self-resonance.

When you connect the loop equivalent circuit to another circuit, such as a transmission line or amplifier, you place a load across the output (AA). A very small impedance load effectively shorts out the capacitance and enables wideband operation using loop current. Alternatively, a very high impedance load provides high output voltages at the resonant frequency, which is how tuned loops work.



Tuned Loop

To make the equivalent circuit into a tuned loop, just add a tuning capacitance across the loop output, and work with open circuit voltages.

$$Q = \frac{\sqrt{F_U F_L}}{F_U - F_L} = \frac{Frequency}{Bandwidth}$$

For example, a tuning capacitor of 10 pF adds to the stray capacitance of the loop and produces resonance at 7.5 MHz with a very sharp Q=120.

Similarly, adding a CT of 100 pF lowers the resonant frequency to 4.0 MHz.



But the bandwidth of the loop becomes quite narrow with a passband near 50 kHz. This means the antenna has to be tuned whenever you change frequency.

Here is the takeaway from all this. When you work with a small loop in Open Circuit Voltage Mode, the antenna <u>must</u> be frequency dependent. This is true because of either a tuning capacitor or stray capacitance. Furthermore, the antenna must be narrowband, with a relatively high Q.

Q is the ratio of reactive to real power in a coil. Voltage across the tuned circuit is multiplied by Q. Voltage magnification takes place at resonance. Thus a very small current can create a very large voltage.

Looking at these numbers, the SPICE model is producing a loop inductance of around 12 μH and capacitance of 30 pF. This produces a self resonance of 8.4 MHz.

Getting a Test Signal

The basic formula for voltage created at the output of an open circuit loop is:

$V = \omega N A \mu_{MAT} H \cos\theta$

This tells us the voltage increases with frequency ($2\pi f$), azimuth of signal (θ) and permeability of loop material (μ). It also increases with loop area (A) and number of turns (N). Finally, the magnetic field strength created is equal to the electric field strength divided by wave impedance ($H_O = E \div 377$).

Since we have a one turn loop (N=1) and can assume maximum azimuth ($\cos\theta = 1$), we can simplify the voltage formula as follows:

$V = \omega A \mu_{MAT} (E \div 377)$.

According to Faraday's law, V_{OC} is driven by the time varying magnetic flux B cutting through the loop.

Permeability is the measure of the ability of a material to support the formation of a magnetic field within itself . 1.26E-6 H/m for aluminum.

So, running these numbers, an electric field strength of 1 µV at 1 MHz into a 1 meter diameter aluminum loop creates an open circuit voltage of 21 pV. Even an S9 signal (50 µV) only creates an open circuit voltage of 1 µV.

So, what practical use can me make of such tiny voltages from a magnetic loop?



$$V = 2\pi\mu_0\ N\ A\ H_0\ f\ \cos\theta$$

Loudet, equation 13, and elsewhere

Voltage versus Current

Even if we could get rid of the stray capacitance, the loop voltage would remain frequency dependent. The resonance peak would be gone, but the gain of the loop would vary with frequency. This does not make for a good wideband antenna, as you would have to continuous compensate for the frequency dependent gain. On the other hand, above a certain point, a short circuit current loop would become frequency independent.

This all comes down to math. If the loop load and loss resistance is very low, they effectively disappear from the equation. Since ω is in the numerator and denominator, it also disappears. You are left with loop current that simply depends on area (A) and inductance (L). Magic!

R_{LOOP} is both radiation resistance and loss resistance. Both R_{RAD} and R_{AC} will change with frequency, while R_{DC} remains constant.

At very low frequencies, current increases while impedance is dominated by resistance. However, when reactance becomes much larger than resistance, current goes flat. The transition frequency for this effect is 3* $F_C = R / 2\pi L$ or in the case of my loop, above 3 * 300 kHz.

In the analysis shown here, R_{LOOP} is given as 0.1Ω, inductance as 2.7μH and R_{LOAD} as 5Ω.



$$I_{SC} = \frac{V}{Z} = \frac{f(\omega BA)}{|(\omega L) + R_{LOOP} + R_{LOAD}|} \approx \frac{f(BA)}{L} \qquad \omega L >> R_{LOAD} > R_{LOSS}$$

Is Loop Current Uniform?

Small loop theory assumes uniform current around the loop.

There has been a lot of controversy about whether or not the assumption that small magnetic loop current is uniform around its circumference. The answer is somewhat.

Where loop shape is less than around 10% wavelength, current is nearly constant, within a few percent. As frequency increases, and the loop size is more like 20-30% wavelength, current is not uniform and can vary as much as ±20%.



So, if you are operating your 1 meter loop at 10 MHz or lower, the uniform current assumption used by many of the analytic formulas pretty much holds.

A more thorough description of loop currents can be found in QEX, Jul/Aug 2018.

Generally, near constant current is achieved with a circumference less than 20% wavelength.

The biggest impact of current variation is in the radiation pattern.

LTSpice Loop Models

- Version B is standalone, the basic loop model
  - Designed for single turn aluminum wire or tube or arbitrary Loop and Conductor diameters, as well as arbitrary Electric Field strength (V/m)
- Version C is modified to be a Hierarchical Block to be used in other LTSpice models
  - Exposes the following parameters: Field Strength (EF V/m), Loop Diameter (LD in meters), Wire/Tube Diameter (WD in meters)

Wideband Magnetic Loop Equivalent Circle Version B (wml_b.asc)



Constants

.param E=1u ; Electric Field V/m
.param Loop_d=1 ; loop diameter in meters
.param Wire_d=14.5m ; wire or tube diameter, meters
.param Mu_mat=1.26e-6 ; loop material permeability for Aluminum H/m
.param Sigma_mat=3.5e7 ; loop material conductivity for aluminum S/m
.param H=({E}/376.73031346177)
.param Loop_r=({Loop_d}/2)
.param Wire_r=({Wire_d}/2)
.param Loop_c=(pi*{Loop_d})
.param DCRes_factor = 0.000021 ; aluminum ohms/meter

Variables

.param Area=pi*{loop_r}**2
.param Len=pi*{loop_d}
.param Lloop= (ln(8*{Loop_r}/{Wire_r})-2)*{Loop_r}*{Mu_mat}
.param Lwire=(2*{Len})*(2.303*log(4*{Len}/{Wire_d})-.75+({Wire_d}/({Len}*2)))*1E-7
.param Rdc=({Loop_c}*{DCRes_factor})
.param Cloop=3.9685E-13*((100*{Loop_C}/Pi)**4/(100*{Loop_C}))**(1/3)

Wideband Magnetic Loop Equivalent Circle Version C (wml_c.asc)
For use only as Hierarchical Block in other LTSpice Models



Constants

.param E={EF}; E=1u ; Electric Field V/m
.param Loop_d={LD}; Loop_d=1 ; loop diameter in meters
.param Wire_d= {WD}; Wire_d=14.5m ; wire or tube diameter, meters
.param Mu_mat=1.26e-6 ; loop material permeability for Aluminum H/m
.param Sigma_mat=3.5e7 ; loop material conductivity for aluminum S/m
.param H=({E}/376.73031346177)
.param Loop_r=({Loop_d}/2)
.param Wire_r=({Wire_d}/2)
.param Loop_c=(pi*{Loop_d})
.param DCRes_factor = 0.000021 ; aluminum ohms/meter

Variables

.param Area=pi*{loop_r}**2
.param Len=pi*{loop_d}
.param Lloop= (ln(8*{Loop_r}/{Wire_r})-2)*{Loop_r}*{Mu_mat}
.param Lwire=(2*{Len})*(2.303*log(4*{Len}/{Wire_d})-.75+({Wire_d}/({Len}*2)))*1E-7
.param Rdc=({Loop_c}*{DCRes_factor})
.param Cloop=3.9685E-13*((100*{Loop_C}/Pi)**4/(100*{Loop_C}))**(1/3)

Wideband Loop Model C (Hierarchical Block) Test Jig

Equivalent Circuit Model, Circular Aluminum Loop

Right-click on wml_c symbol
to view Equivalent Circuit



EF=1u LD=1 WD= 14.5m

Inputs:
EF electric field signal received V/m
LD circular loop diameter, meters
WD loop wire/tube diameter, meters

Analysis: Probe V(out) to see voltage from loop
Probe I(RLoad) to see current in loop

.ac dec 200 1e4 4e7
.step param Rload list 1m 50 500k
.param RLoad=1

1m load simulates loop short circuit (work with current)
500k load simulates loop open circuit (work with voltage)

Generating a Signal

1. Select Electric Field Strength E. Default is 1 µV/m.
2. Convert to Magnetic Field Strength H by dividing by 377. Generate this field at node labeled H.
3. Use a Behavioral Voltage Source, Bloop, to implement V=ωµAH to obtain a frequency dependent signal voltage across the AC sweep range.

Since LTSpice does not provide direct access to sweep frequency, we need to use a Laplace Transfer Function in the B source to obtain frequency from the “s domain”.

The Bloop formula requires access to a node, so we create node labelled “H” as the magnetic field strength.




```
.param E={EF}; E=1u ; Electric Field V/m
.param H=({E}/376.73031346177)

.param Loop_r=({Loop_d}/2)
.param Area=pi*{loop_r}**2

.param Mu_mat=1.26e-6  ; loop material permeability for Aluminum H/m
```

Static Parameter R_{dc} – DC Loss

To keep this simple, the model contains the resistive loss per meter for aluminum tube , which is labelled DCRes_factor.

DC Loss resistance is simply calculated as the length of the loop (circumference, or Loop_c) multiplied by resistance of 21 $\mu\Omega$ per meter.




```
.param Rdc=({Loop_c}*{DCRes_factor})
.param Loop_c=(pi*{Loop_d})
```

```
.param DCRes_factor =  0.000021 ; aluminum ohms/meter
```

Static Variables - Inductance

This is comprised of Loop Inductance (Lloop) and Wire Inductance (Lwire).

I found various formulas for circular loop inductance, tested them in an Excel spreadsheet for consistence, and also checked against various inductance calculators on the web. The key variables are material permeability, loop radius and wire radius.

Similarly, there are several available formulas for inductance of a straight conductor. I think these formulas overstate the added wire inductance from an aluminum loop, but for now I left this effect in the model until I learn more.




```
.param Len=pi*{loop_d}
.param Lloop= (ln(8*{Loop_r}/{Wire_r})-2)*{Loop_r}*{Mu_mat}
.param Lwire=(2*{Len})*(2.303*log(4*{Len}/{Wire_d})-.75+({Wire_d}/({Len}*2)))*1E-7
```

$$L_{circle} \approx N^2 R \mu_0 \mu_r \left[\ln\left(\frac{8R}{a}\right) - 2 \right]$$

https://technick.net/tools/inductance-calculator/circular-loop/

$$L = 2l\left[2.303 \log(4l/d) - 1 + \mu/4 + (d/2l)\right]$$

http://www.consultrsr.net/resources/eis/induct5.htm

Static Variables – Stray Capacitance Cloop

Capacitance is a measure of the ability of a conductor or a system of conductors to store charge (and hence to store energy). With a multi-turn coil or loop, there is a lot of capacitance between the turns. For a single conductor, capacitance characterizes the ability of the conductor to accumulate an electrical charge.

We are then considering intrinsic rather than mutual (partial) capacitance. The second (virtual) conductor is the sphere (enclosing volume) having zero potential.



$$C_{\text{loop}} \approx 3.9685 \cdot 10^{-13} \cdot \sqrt[3]{\frac{\left(\frac{400\,w}{\pi}\right)^4}{100\,l}}$$

Loudet, Equation 20

```
.param Cloop=3.9685E-13*((100*{Loop_C}/Pi)**4/(100*{Loop_C}) )**(1/3)
```

Dynamic Variables – Radiation Resistance R_{RAD}

Radiation Resistance varies by frequency, and for loop antennas is very low.

Modeling RRAD was implemented as an LTSpice sub-circuit. The input parameter is loop Area, which defaults to 0.785 square meters.

The effect was modelled with a voltage dependent current source, G1.

I implemented the sub circuit using the Trask equation, validated against Loudet.



$$R_{RAD} = 320\,\pi^4\,(A / \lambda^2)^2\,N^2$$

Trask, QEX Jul/Aug 2003, equation 1

$$R_{rad} = \frac{32\,\pi^4\,10^{-7}}{3\,c^3}(N\,\mu_r\,A)^2\,f^4 = \frac{2 \cdot 10^{-7}}{3\,c^3}(N\,\mu_r\,A)^2\,\omega^4$$

Loudet, equation 15

```
* John's Radiation Resistance
.SUBCKT RRAD 1 2 PARAMS: A=1
G1 1 2 1 2 laplace=( 1.25k * {p}* 1/({a}/(({k}/(abs(s)/{twopi}))^2)^2))
R 1 2 10meg
.param k=299792
.param twopi = 2*{pi}
.param p=320*{pi}**4
.ends
```

Dynamic Variables – AC Loss R_{AC}

AC loss is frequency dependent and depends on loop perimeter (Len), wire diameter (Wire_d), as well as material permittivity and permeability (Sigma_mat and Mu_mat). These are required to calculate skin effect .

The sub circuit implements the Trask equations and validated results against Loudet.



L={Len} d={Wire_d} Sigma={Sigma_mat} Mu={Mu_mat}

.param Mu_mat=1.26e-6 ; loop material permeability for Aluminum H/m
.param Sigma_mat=3.5e7 ; loop material conductivity for aluminum S/m

$$R_{ac} = \frac{4\,N\,w}{\pi\,d}\sqrt{\pi\,\mu_0\,f\,\rho}$$

Loudet, equation 19

$$R_{loss} = \frac{L}{\sigma\,\pi\,d\,\delta} = \frac{L}{d}\sqrt{\frac{f\,\mu_0}{\pi\,\sigma}} \qquad \delta = \frac{1}{\sqrt{f\,\pi\mu\sigma}}$$

Trask, QEX Jul/Aug 2003 equations 2 and 3

- John's Skin Effect Loss

```
.SUBCKT RAC 1 2 PARAMS: L=1 d=1 Sigma=1 Mu=1
G1 1 2 1 2 laplace={L}/({pre}*(1/sqrt(1/((abs(s)/{twopi}))*{post}))*2e-10)
R 1 2 10meg
* The 2e-10 is kluge to get fit.
.param twopi = 2*pi
.param pre=pi*{d}*{Sigma}
.param post=pi*{Mu}*{Sigma}
.ends
```

Credits

Please see the References on the following page.

The Loudet model was my inspiration and major source of learning. Since he did not provide his sub-circuits for R_{RAD} and R_{AC}, I had to develop those myself. Each equation was generally checked against alternative forms and online calculators to ensure consistency of results.

This is the first significant modeling I have ever done with LTSpice, so mistakes are quite possible. I do not claim any expertise in the field. Best efforts only!

☺

References

Web

Many articles on active antennas and wideband loops on the web. These are the sources I found most helpful in developing my understanding, particularly Loudet.

Bruno, Thinking About Ideal Loops, http://www.vlf.it/looptheo7/looptheo7.htm

Bauer, Broadband Active Loop Antennas, https://www.radiomuseum.org/forum/broadband_active_loop_antennas_1.html

Levkov at http://www.lz1aq.signacor.com/ (many articles)

Morgan, Analysis and Calibration of Loop Probes for use in Measuring Interference Fields, Naval Research Laboratory 1949 http://www.dtic.mil/dtic/tr/fulltext/u2/b204978.pdf (appendices contain many loop formulas)

Trask, Active Loop Antennas for HF Reception, Parts 1 and 2, QEX 2003.

Loop Antenna LTSpice Model at http://shodhganga.inflibnet.ac.in/bitstream/10603/116507/11/11_chapter%204.pdf

Loudet, Magnetic Loop Antenna Theory https://sidstation.loudet.org/antenna-theory-en.xhtml

Textbooks

Kraus, Antennas, Second Edition, McGaw-Hill, 1988

Survey of Small Antenna Theory, McGraw-Hill Professional

Fujimoto and Morishita, Modern Small Antennas, Cambridge University Press, 2013